Racionális törtfüggvények integrálása

Racionális függvények (ismétlés)

Polinomok

A valós egyváltozós polinomfüggvény definiciója :

```
> restart:
  poli:=x->Sum(a[i]*x^i,i=0..n)=sum(a[i]*x^i,i=0..n);     #poli:
  R->R
```

$$poli := x \rightarrow \sum_{i=0}^{n} a_i x^i = \sum_{i=0}^{n} a_i x^i \qquad (1.1.1)$$

```
> n:=5:#Legyen pl. n=5
> poli(x);
```

$$\sum_{i=0}^{5} a_i x^i = a_0 + a_1 x + a_2 x^2 + a_3 x^3 + a_4 x^4 + a_5 x^5 \qquad (1.1.2)$$

Tekintsük át a polinomok kezelésével kapcsolatos fontosabb Maple utasításokat! Vegyük a következő polinomot:

```
> poli1:=x^6-x^5-9*x^4+x^3+20*x^2+12*x;
```

$$poli1 := x^6 - x^5 - 9x^4 + x^3 + 20x^2 + 12x \qquad (1.1.3)$$

A factor utasítás szorzattá alakítja :

```
> factor(poli1);
```

$$x\,(x-2)\,(x-3)\,(x+2)\,(x+1)^2 \qquad (1.1.4)$$

```
> poli2:=(x+3);
```

$$poli2 := x + 3 \qquad (1.1.5)$$

Az expand ellentétes hatású, kifejti a polinomot (kifejezést):

```
> poli3:=expand(poli2^6);
```

$$poli3 := x^6 + 18x^5 + 135x^4 + 540x^3 + 1215x^2 + 1458x + 729 \qquad (1.1.6)$$

```
> factor(poli3);
```

$$(x+3)^6 \qquad (1.1.7)$$

A solve megkeresi a zérushelyeket:

```
> poli1;
```

$$x^6 - x^5 - 9x^4 + x^3 + 20x^2 + 12x \qquad (1.1.8)$$

```
> `zérushelyek`:=solve(poli1,x);
```

$$zérushelyek := 0, -2, 2, 3, -1, -1 \tag{1.1.9}$$

A zérushelyek értékek listáját alkotják ,a lista elemeit elérhetjük:

```
> x2:=`zérushelyek`[3];
```

$$x2 := 2 \tag{1.1.10}$$

A faktor különbözõ második argumenttel más-más számtest felett alakítja szorzattá a polinomot:

```
> poli4:=x^4-x^2-2;
```

$$poli4 := x^4 - x^2 - 2 \tag{1.1.11}$$

```
> factor(poli4);
```

$$(x^2 - 2)(x^2 + 1) \tag{1.1.12}$$

```
> factor(poli4,real);
```

$$(x + 1.414213562)(x - 1.414213562)(x^2 + 1.) \tag{1.1.13}$$

```
> factor(poli4,sqrt(2));
```

$$(x^2 + 1)(x + \sqrt{2})(x - \sqrt{2}) \tag{1.1.14}$$

```
> factor(poli4,complex);
```

$$(x + 1.414213562)(x + 1.\,I)(x - 1.\,I)(x - 1.414213562) \tag{1.1.15}$$

```
> with(plots):
Warning, the name changecoords has been redefined
```

Ábrázoljuk a poli1 polinomot!

```
> T:=textplot([-1.2,-15,`Poli1=`],align=LEFT):
> P:=plot(poli1(x),x=-2.2..3.05):
> afug:=convert(poli1,string):
> H:=textplot([-1.2,-15,afug],align=RIGHT):
> display([P,T,H],title=`a poli1 polinom grafikonja`);
```



>

Figyeljük meg ,az $x = -1$ *kétszeres zérushelye a függvénynek, az (x+1) gyöktényezõ második hatványon szerepel.Itt érinti a görbe az x-tengelyt!*

```
> poli5:=x-3*x^4+5*x^3-1-7*x^2;
```

$$poli5 := x - 3x^4 + 5x^3 - 1 - 7x^2 \qquad (1.1.16)$$

A degree utasítás megadja a polinom fokszámát:

```
> `fokszám`=degree(poli5);
```

$$fokszám = 4 \qquad (1.1.17)$$

A" sort " rendezi - csökkenõleg - a polinomot:

```
> poli5:=sort(poli5,x);
```

$$poli5 := -3x^4 + 5x^3 - 7x^2 + x - 1 \qquad (1.1.18)$$

Megkérdezhetjük a polinom adott fokszámú tagjának együtthatóját:

```
> coeff(poli5,x,0);
```

$$-1 \qquad (1.1.19)$$

Az e egy lista ,amely a polinom együtthatóit adja meg, növekvõ fokszám szerint rendezve:

```
> e:=[seq(coeff(poli5,x,i),i=0..degree(poli5))];
```

$$e := [-1, 1, -7, 5, -3] \tag{1.1.20}$$

```
> e[3];
```

$$-7 \tag{1.1.21}$$

Ha csak az együtthatókra vagyunk kíváncsiak, akkor a coeffs választ ad:

```
> `együtthatók`:=coeffs(poli5);
```

$$\mathit{együtthatók} := -1, 1, -7, 5, -3 \tag{1.1.22}$$

Feladat

Határozzuk meg azt a legfeljebb harmadfokú polinomfüggvényt, amely áthalad a következõ pontokon: A(1,6); B(2,-4); C(-2,0); D(5,14).

```
> restart:
```

Létrehozzuk a polinomfüggvényt, majd a négy egyenletet:

```
> f:=x->a[3]*x^3+a[2]*x^2+a[1]*x+a[0];
```

$$f := x \rightarrow a_3 x^3 + a_2 x^2 + a_1 x + a_0 \tag{1.1.23}$$

```
> e1:=subs(x=1,f(x))=6;
> e2:=subs(x=2,f(x)=-4);
> e3:=subs(x=-2,f(x))=0;
> e4:=subs(x=5,f(x))=14;
```

$$e1 := a_3 + a_2 + a_1 + a_0 = 6$$

$$e2 := 8\,a_3 + 4\,a_2 + 2\,a_1 + a_0 = -4$$

$$e3 := -8\,a_3 + 4\,a_2 - 2\,a_1 + a_0 = 0$$

$$e4 := 125\,a_3 + 25\,a_2 + 5\,a_1 + a_0 = 14 \tag{1.1.24}$$

Megoldjuk az egyenletrendszert, majd az assign hatására a változók felveszik a kapott értékeket, az unapply pedig létrehozza a konkrét függvényt:

```
> S:=solve({e1,e2,e3,e4},{a[1],a[2],a[3],a[0]});
```

$$S := \{a_1 = -5, a_2 = -4, a_3 = 1, a_0 = 14\} \tag{1.1.25}$$

```
> assign(S);
> f(x);
```

$$x^3 - 4\,x^2 - 5\,x + 14 \tag{1.1.26}$$

```
> f:=unapply(f(x),x);
```

$$f := x \rightarrow x^3 - 4\,x^2 - 5\,x + 14 \tag{1.1.27}$$

Keresük meg a függvény zérushelyeit!

```
> solve(f(x),{x});
```

$$\{x = -2\}, \{x = 3 + \sqrt{2}\}, \{x = 3 - \sqrt{2}\} \tag{1.1.28}$$

```
> `a_zérushelyek`:=evalf(%);
```

$$\mathit{a\_zérushelyek} := \{x = -2.\}, \{x = 4.414213562\}, \{x = 1.585786438\} \tag{1.1.29}$$

Végül ábrázoljuk a függvényt!

```
> with(plots):
Warning, the name changecoords has been redefined
> Kep:=plot(f(x),x=-2.4..4.6,title=`A polinomfüggvény`):
> szoveg:=textplot([1,2,`f(x)=`],align=LEFT):
> fuggveny:=textplot([1,2,convert(f(x),string)],align=RIGHT):
> display({Kep,szoveg,fuggveny});
```



Próbáljunk más-más pontokat felvenni! Azt tapasztaljuk, hogy négy pont mindig egyértelmûen meghatároz egy legfeljebb harmadfokú polinomfüggvényt!
Általában n+1 pont egyértelmûen meghatároz egy legfeljebb n-edfokú polinomot! Szemlélgessük ezt a tényt különbözõ számú pont ill. polinom felvételével!

Például:
Határozzuk meg a következõ pontokra illeszkedõ, legfeljebb negyedfokú polinomot: A(1,3), B(-2,5), C(0,4), D(-4,-3), E(5,8)!

Láttuk, hogy ha a polinomnak többszörös gyöke van, akkor a gyök környezetében jellegzetesen viselkedik a polinom:görbéje érinti az x tengelyt. Vizsgáljuk meg ezt a jelenséget egy kicsit részletesebben, animáció segítségével:

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
> Trafo:=proc(f,x0,a,b,c,d,N,Sx,Sy)
>   local H,n,szam,S,P:
>   H:=textplot([Sx,Sy,`f(x)=`],align=LEFT):
>   for n from 1 to N  do
>     szam:=convert(1/3*f(x)*(x-x0)^n,string):
>     S:=textplot([Sx,Sy,szam],align=RIGHT):
>     P:=plot(1/3*f(x)*(x-x0)^n,x=a..b,y=c..d,scaling=
  constrained,color=BLUE,thickness=2):
>     Kep||n:=display([P,H,S]):
>   od:
> end:
> f:=x->x+1;N:=8:
```

$$f := x \rightarrow x + 1 \qquad (1.1.30)$$

```
> Trafo(f,1,-2.4,3,-6,6,8,3,0.85):
> display([Kep||(1..N)],title=`Többszörös gyök`,insequence=true,
  scaling=constrained);
```



Racionális törtfüggvények

Definíció

Két polinom hányadosát racionális törtfüggvénynek nevezzük.

```
> f:=x->(x^4-3*x^2+5*x-4)/(x^3-6*x^2+5*x-2);
```

$$f := x \rightarrow \frac{x^4 - 3x^2 + 5x - 4}{x^3 - 6x^2 + 5x - 2} \qquad (1.2.1)$$

Def.:A racionális törtet áltörtnek nevezzük ha a számláló fokszáma legalább akkora, mint a nevezõjé,egyébként valódi törtrõl beszélünk.

Az áltörteket - polinomosztás segítségével - átalakíthatjuk polinom és valódi tört összegére:

```
> `tört`:=f(x);
```

$$tört := \frac{x^4 - 3x^2 + 5x - 4}{x^3 - 6x^2 + 5x - 2} \qquad (1.2.2)$$

A tört számlálóját ill. nevezõjét a numer(kifejezés) ill a denom(kifejezés) utasítással kaphatjuk meg:

```
> `számláló`:=numer(`tört`);
```

$$számláló := x^4 - 3x^2 + 5x - 4 \tag{1.2.3}$$

```
> `nevezõ`:=denom(`tört`);
```

$$nevezõ := x^3 - 6x^2 + 5x - 2 \tag{1.2.4}$$

A következõ sorban láthatjuk, hogyan történik a polinomosztás a quo utasítással. A negyedik argumentumként megadott r a maradékot is megadja:

```
> `hányados`:=quo(`számláló`,`nevezõ`,x,'r');
```

$$hányados := x + 6 \tag{1.2.5}$$

```
> `maradék`:=sort(r);#Rendezve kapjuk meg a maradékot#
```

$$maradék := 28x^2 - 23x + 8 \tag{1.2.6}$$

```
> `tört`:=`hányados`+`maradék`/`nevezõ`;
```

$$tört := x + 6 + \frac{28x^2 - 23x + 8}{x^3 - 6x^2 + 5x - 2} \tag{1.2.7}$$

```
> `tört`:=normal(`tört`);
```

$$tört := \frac{x^4 - 3x^2 + 5x - 4}{x^3 - 6x^2 + 5x - 2} \tag{1.2.8}$$

Nézzük meg, van e olyan valós x érték, melyre a tört nem értelmezhetõ:

```
> `A nevezõ zérushelyei`:=fsolve(`nevezõ`,x);
```

$$A\ nevezõ\ zérushelyei := 5.095823999 \tag{1.2.9}$$

Alakítsuk szorzattá - a valós számok halmazán -a nevezõt:

```
> factor(`nevezõ`,real);
```

$$(x - 5.095823999)\left(x^2 - 0.9041760014\,x + 0.3924782333\right) \tag{1.2.10}$$

Láthatóan a második tényezõnek már nincsenek valós gyökei. Ezt mutatja a nevezõ grafikonja is:

```
> plot(`nevezõ`,x=-2..6,title=`A nevezõ`);
```




```
> expr:=(x-1)^2/(x^2-1);
```

$$expr := \frac{(x-1)^2}{x^2-1} \qquad (1.2.11)$$

```
> discont(expr,x);
```

$$\{-1, 1\} \qquad (1.2.12)$$

```
> plot(expr,x=-5..5,y=-10..10);
```




```
> kif:=x*(x+4)*(x-3)/(x^2-1);
```

$$kif := \frac{x\,(x+4)\,(x-3)}{x^2-1} \tag{1.2.13}$$

```
> discont(%,x);
```

$$\{-1, 1\} \tag{1.2.14}$$

```
> aszimptota:=quo(numer(kif),denom(kif),x,'r');
```

$$aszimptota := x+1 \tag{1.2.15}$$

```
> plot({kif,aszimptota},x=-20..20,y=-15..15,discont=true,title=
  `Törtfüggvény és aszimptotája`);
```



A következõ animáció azt szemlélteti, hogyan változik a számláló fokszámának növelésével a racionális törtfüggvény aszimptotája:

```
> restart:
> with(plots):
Warning, the name changecoords has been redefined
> Asimptotak:=proc(f,a,b,c,d,N,Sx,Sy)
>   local H,n,szam,S,P,g,aszimptota:
>   H:=textplot([Sx,Sy,`f(x)=`],align=LEFT):
>   for n from 1 to N  do
>     szam:=convert((f(x)^n+1)/x,string):
>     S:=textplot([Sx,Sy,szam],align=RIGHT):
>     g:=(f(x)^n+1)/x:
>     aszimptota:=quo(numer(g),denom(g),x,'r'):
>     P:=plot({g,aszimptota},x=a..b,y=c..d,discont=true,scaling=
  constrained):
>     Kep||n:=display([P,H,S]):
```

```
> od:
> end:
> f:=x->x;N:=6;
```

$$f := x \rightarrow x$$

$$N := 6 \tag{1.2.16}$$

```
> Asimptotak(f,-5,5,-4,6,N,3.4,4.5):
> display([Kep||(1..N)],title=`Függvény és aszimptota`,
  insequence=true);
```



Racionális törtfüggvények integrálása

Mindenekelõtt elevenítsük fel a racionális törtfüggvény fogalmát! Nézzünk elõször néhány példát!

```
> restart:
> f:=x-> (x^3+3*x^2-4*x+4)/(x^2-5*x-4);
```

$$f := x \rightarrow \frac{x^3 + 3x^2 - 4x + 4}{x^2 - 5x - 4} \tag{2.1}$$

```
> g:=x->(x-2)/(x^2+4*x+5);
```

$$g := x \to \frac{x-2}{x^2+4x+5} \tag{2.2}$$

Definíció

A $\frac{p(x)}{q(x)}$ alakú függvényt, ahol p(x) és q(x) polinom, racionális törtfüggvénynek nevezzük.
Ha a nevezõben szereplõ polinom fokszáma nagyobb a számlálóban levõnél, akkor **valódi tört**rõl, egyébként **áltört**rõl beszélünk.

1. Lépés: Fokszámvizsgálat, szükség esetén polinomosztás

Minden *áltört* felbontható egy *polinom* és egy *valódi tört* összegére. Mivel a polinomfüggvény integrálása nem okoz problémát, ezért a valódi törtek integrálásával kell csak foglalkoznunk.
A polinom és valódi tört össszegére való felbontást **polinomok maradékos osztásá**val végezzük.
Tekintsük a következõ törtet:

```
> f:=x->(x^5-x^4-3*x+5)/(x^4-2*x^3+2*x^2-2*x+1);
```

$$f := x \to \frac{x^5-x^4-3x+5}{x^4-2x^3+2x^2-2x+1} \tag{2.1.1}$$

A \`számláló\` ill. \`nevezõ\` nevû változókban tároljuk a törtet alkotó polinomokat:

```
> `számláló`:=x^5-x^4-3*x+5;
```

$$számláló := x^5-x^4-3x+5 \tag{2.1.2}$$

```
> `nevezõ`:=x^4-2*x^3+2*x^2-2*x+1;
```

$$nevezõ := x^4-2x^3+2x^2-2x+1 \tag{2.1.3}$$

A hányados:

```
> Q:=quo(`számláló`,`nevezõ`,x);
```

$$Q := x+1 \tag{2.1.4}$$

A maradék:

```
> R:=rem(`számláló`,`nevezõ`,x);
```

$$R := 4-2x \tag{2.1.5}$$

Tört=Hányados + Maradék/Osztó:

```
> `f(x)`:=f=Q+R/`nevezõ`;
```

$$f(x) := f = x+1+\frac{4-2x}{x^4-2x^3+2x^2-2x+1} \tag{2.1.6}$$

2. Lépés: A nevezõ vizsgálata, szorzattá alakítása

A valódi törtet egyszerûbb, úgynevezett elemi törtek összegére szeretnénk felbontani, ezeket ugyanis már tudjuk integrálni. Ehhez elõször a tört nevezõjét kell - általában - szorzattá alakítani.
Ennek elvi lehetõségét biztosítja a következõ tétel.
TÉTEL. Minden polinom felbontható valós együtthatós elsõ- és másodfokú polinomok szorzatára.
A felbontás nem mindig egyszerû. A Maple segítségével azonban általában könnyen elvégezhetjük. A factor utasítást használjuk, szükség esetén kiegészítõ opciót adunk. Pl.:

```
> `nevezõ`;
```

$$x^4-2x^3+2x^2-2x+1 \tag{2.2.1}$$

```
> factor(`nevezõ`);
```

$$\left(x^2+1\right)\left(x-1\right)^2 \tag{2.2.2}$$

```
> factor(x^6+1);
```

$$\left(x^2+1\right)\left(x^4-x^2+1\right) \tag{2.2.3}$$

```
> factor(op(2,%),sqrt(3));
```

$$\left(x^2-x\sqrt{3}+1\right)\left(x^2+x\sqrt{3}+1\right) \tag{2.2.4}$$

A valós számok halmaza felett szorzattá alakítva:

```
> factor(x^6+x^4+5,real);
```

$$\left(x^2+2.047064056\,x+1.537063974\right)\left(x^2+2.116343297\right)\left(x^2-2.047064056\,x+1.537063974\right) \tag{2.2.5}$$

A komplex számok halmazán szorzattá alakítva:

```
> factor(%,complex);
```

$$(x+1.023532028+0.6996042894\,\mathrm{I})\,(x+1.023532028-0.6996042894\,\mathrm{I})\,(x+1.454765719\,\mathrm{I})\,(x-1.454765719\,\mathrm{I})\,(x-1.023532028+0.6996042894\,\mathrm{I})\,(x-1.023532028-0.6996042894\,\mathrm{I}) \tag{2.2.6}$$

3. Lépés. A valódi törtet résztörtek összegére bontjuk

A résztörtek alakja a nevezõ tényezõitõl függ, a következõképpen:
a) Legyen **x-r** a nevezõ **lineáris tényezõ**je (valós gyöktényezõ), s tartalmazza a nevezõ ezt a tényezõt m-edik hatányon, azaz legyen $(x-r)^m$ az a legnagyobb hatványa a tényezõnek, amellyel a nevezõ osztható. Ekkor a következõ törtek jönnek létre a felbontáskor:

$$\frac{A_1}{x-r}+\frac{A_2}{(x-r)^2}+\ldots+\frac{A_m}{(x-r)^m}.$$

Ugyanez igaz, a megfelelõ kitevõvel valamennyi elsõfokú, tehát valós ,$(x-x_0)$ alakú gyöktényezõre.
b) Legyen $x^2+p\,x+q$ nevezõ **négyzetes**, tehát komplex gyöktényezõje, ahol $p^2-4\,q<0$.
Tegyük fel, hogy a nevezõ n-edik hatványon tartalmazza a kérdéses tényezõt, azaz

$$\left(x^2+p\,x+q\right)^n$$

emelhetõ ki maximálisan a nevezõbõl. Ekkor ezen tényezõ a következõ törteket indukálja:

$$\frac{B_1x+C_1}{x^2+p\,x+q}+\frac{B_2x+C_2}{\left(x^2+p\,x+q\right)^2}+\ldots+\frac{B_nx+C_n}{\left(x^2+p\,x+q\right)^n}.$$

Ugyanez igaz, a megfelelõ kitevõvel, minden tényezõre.
Természetesen az eredeti tört a kapott résztörtek összegével egyenlõ.

4.Lépés. Meghatározzuk a résztörtek számlálóiban szereplõ együtthatókat.

A résztörteket közös nevezõre hozzuk, s a számlálóban kapott polinom együtthatóit összehasonlítjuk az eredeti tört számlálójában levõ polinom együtthatóival. Az így adódó egyenletrendszert megoldva kapjuk a keresett együtthatókat

Példa. Bontsuk résztörtekre az

$$\frac{f(x)}{g(x)} = \frac{-2x+4}{(x-1)^2\left(x^2+1\right)}$$

```
> `valódi tört`:=R/factor(`nevezõ`);
```

$$\textit{valódi tört} := \frac{4-2x}{\left(x^2+1\right)(x-1)^2} \tag{2.4.1}$$

A nevezõ felbontása alapján a résztörtek:

```
> `résztörtek`:=(A*x+B)/(x^2+1)+C/(x-1)+D/(x-1)^2;
```

$$\textit{résztörtek} := \frac{Ax+B}{x^2+1} + \frac{C}{x-1} + \frac{\mathrm{D}}{(x-1)^2} \tag{2.4.2}$$

```
> `valódi tört`-`résztörtek`=0;
```

$$\frac{4-2x}{\left(x^2+1\right)(x-1)^2} - \frac{Ax+B}{x^2+1} - \frac{C}{x-1} - \frac{\mathrm{D}}{(x-1)^2} = 0 \tag{2.4.3}$$

A törteket közös nevezõre hozzuk:

```
> normal(lhs(%));
```

$$-\frac{1}{\left(x^2+1\right)(x-1)^2}\left(-4+2x+Ax^3-2Ax^2+Ax+Bx^2-2Bx+B+Cx^3-Cx^2+Cx-C+\mathrm{D}x^2+\mathrm{D}\right) \tag{2.4.4}$$

Az x hatványai szerint rendezünk:

```
> collect(numer(%),x);
```

$$(-A-C)x^3 + (-\mathrm{D}-B+2A+C)x^2 + (2B-A-2-C)x + 4-\mathrm{D}-B+C \tag{2.4.5}$$

Felírjuk az egyenletrendszert.(Valamennyi egyenlet jobb oldalán 0 áll)

```
> `egyenletrendszer`:={coeffs(%,x)};
```

$$\textit{egyenletrendszer} := \{-A-C,\ -\mathrm{D}-B+2A+C,\ 4-\mathrm{D}-B+C,\ 2B-A-2-C\} \tag{2.4.6}$$

Megoldjuk az egyenletrendszert :

```
> `megoldás`:=solve(`egyenletrendszer`);
```

$$\textit{megoldás} := \{\mathrm{D}=1,\ C=-2,\ B=1,\ A=2\} \tag{2.4.7}$$

A megoldást behelyettesítjük a résztörtekbe:

```
> `parciális törtek`:=subs(`megoldás`,`résztörtek`);
```

$$\textit{parciális törtek} := \frac{2x+1}{x^2+1} - \frac{2}{x-1} + \frac{1}{(x-1)^2} \tag{2.4.8}$$

5. Lépés Integráljuk a kapott törteket.

Integráljuk a parciális törteket!

```
> map(Int,`parciális törtek`,x);
```

$$\int \frac{2x+1}{x^2+1}\,dx + \int -\frac{2}{x-1}\,dx + \int \frac{1}{(x-1)^2}\,dx \tag{2.5.1}$$

```
> value(%);
```

$$\ln(x^2+1) + \arctan(x) - 2\ln(x-1) - \frac{1}{x-1} \tag{2.5.2}$$

A Maple persze mindezt egy szuszra is elvégzi:

```
> Int(f(x),x)=int(f(x),x);
```

$$\int \frac{x^5-x^4-3x+5}{x^4-2x^3+2x^2-2x+1}\,dx = \frac{1}{2}x^2 + x + \ln(x^2+1) + \arctan(x) - 2\ln(x-1) - \frac{1}{x-1} \tag{2.5.3}$$

6.Lépés A résztörtek integrálásáról bõvebben

A racionális törtfüggvények mindig integrálhatók. Ez azt jelenti, hogy az elõzõkben leírt felbontás mindig létezik, s így a következõ típusú *résztörtek* (parciális törtek) *integrálása* végzendõ el:

$$\alpha)\quad \int \frac{1}{(x-r)^h}\,dx\,; \qquad \beta)\quad \int \frac{ax+b}{x^2+px+q}\,dx$$

Az elsõ típus probléma mentesen "elintézhetõ". Ha $h \neq -1$, akkor:

```
> Int(1/(x-r)^h,x)=simplify(int(1/(x-r)^h,x));
```

$$\int \frac{1}{(x-r)^h}\,dx = -\frac{(x-r)^{(-h+1)}}{h-1} \tag{2.6.1}$$

Például:

```
> Int(1/((x-3)^4),x)=int(1/((x-3)^4),x);
```

$$\int \frac{1}{(x-3)^4}\,dx = -\frac{1}{3}\,\frac{1}{(x-3)^3} \tag{2.6.2}$$

Ha h = -1, akkor

```
> Int(1/(x-r),x)=int(1/(x-r),x);
```

(2.6.3)

$$\int \frac{1}{x-r}\,dx = \ln(x-r) \qquad (2.6.3)$$

Az utóbbinak csak speciális esetével foglalkozunk. Itt a nevezõnek nincsenek valós zérushelyei, s az általunk vizsgált esetben másodfokú!

```
> Int((3*x+4)/(x^2+4*x+5),x)=int((3*x+4)/(x^2+4*x+5),x);
```

$$\int \frac{3x+4}{x^2+4x+5}\,dx = \frac{3}{2}\ln\left(x^2+4x+5\right) - 2\arctan(x+2) \qquad (2.6.4)$$

Nézzük, mi is történt! A számlálóban kialakítottuk a nevezõ deriváltját, s úgy csináltunk két törtet, hogy a második számlálója állandó legyen:

```
> Int((3*x+4)/(x^2+4*x+5),x)=3/2*Int((2*x+4)/(x^2+4*x+5),x)-2*
  Int(1/(x^2+2*x+5),x);
```

$$\int \frac{3x+4}{x^2+4x+5}\,dx = \frac{3}{2}\int \frac{2x+4}{x^2+4x+5}\,dx - 2\int \frac{1}{x^2+2x+5}\,dx \qquad (2.6.5)$$

A második integrál nevezõjét $(x+2)^2+1$ alakban írva már érthetõ az eredmény:

```
> Int((3*x+4)/(x^2+4*x+5),x)=int((3*x+4)/(x^2+4*x+5),x);
```

$$\int \frac{3x+4}{x^2+4x+5}\,dx = \frac{3}{2}\ln\left(x^2+4x+5\right) - 2\arctan(x+2) \qquad (2.6.6)$$

A következõ példa azt illusztrálja, hogy mekkora - szinte elvégezhetetlen munkától ment meg bennünket a Maple! (S itt a részletszámítások oldalakat tennének ki!)

```
> Int((4*x^3-5*x^2+7*x-6)/((x^2+3*x+10)^2*(x^2+2*x+5)^4*(x-2)^3*
  (x+1)^4),x)=int((4*x^3-5*x^2+7*x-6)/((x^2+3*x+10)^2*(x^2+2*
  x+5)^4*(x-2)^3*(x+1)^4),x);
```

$$\int \frac{4x^3-5x^2+7x-6}{\left(x^2+3x+10\right)^2\left(x^2+2x+5\right)^4(-2+x)^3(x+1)^4}\,dx = \frac{25}{1769472}\,\frac{1}{(x+1)^2} \qquad (2.6.7)$$

$$+ \frac{1246717}{28149950088000}\ln(-2+x) + \frac{1}{29246464000}\,\frac{2243012\,x+5205324}{\left(x^2+2x+5\right)^2}$$

$$+ \frac{2926093}{233971712000}\,\frac{2x+2}{x^2+2x+5} + \frac{43275205501}{988530483200000}\arctan\left(\frac{1}{2}x+\frac{1}{2}\right)$$

$$+ \frac{21533}{1638400000}\ln\left(x^2+3x+10\right) - \frac{467501}{787251200000}\sqrt{31}\arctan\left(\frac{1}{31}\right.(2x$$

$$+ 3) \sqrt{31} \Big) + \frac{1}{168729600} \frac{-5396 x + 42948}{\left(x^2 + 2 x + 5\right)^3} - \frac{1349}{269967360} \frac{2 x + 2}{\left(x^2 + 2 x + 5\right)^2}$$

$$- \frac{1997766799}{61783155200000} \ln\left(x^2 + 2 x + 5\right) + \frac{2117}{42467328} \frac{1}{x + 1}$$

$$+ \frac{1}{10158080000} \frac{-21419 x - 131902}{x^2 + 3 x + 10} + \frac{1661}{36089679600} \frac{1}{-2 + x}$$

$$+ \frac{1}{1901020160000} \frac{-14180924 x + 43009132}{x^2 + 2 x + 5} - \frac{11}{663552} \frac{1}{(x + 1)^3}$$

$$- \frac{1}{92537640} \frac{1}{(-2 + x)^2} + \frac{7327}{191102976} \ln(x + 1)$$

>

>